Natural Science Report, Ochanomizu University, Vol. 43, No. 1, 1992
お茶の水女子大学 自然科学報告 第43巻 第1号

Fuzzy Temperature Control Programs for the Precision Water Bath of the Heat Exchange Calorimeter

Shuko Fujieda and Shiho Tanaka

Department of Chemistry, Faculty of Science,
Ochanomizu University, Bunkyo-ku, Tokyo

(Received April 30, 1992)

Abstract

As a heat exchange calorimeter may be usually operated at a non-air-conditioned laboratory, the precision water bath is needed in practice. For the reason, it was necessary that the thermal fluctuation of the bath water was kept within a narrow temperature range. In this work, fuzzy inference approach was applied to select the control input for the water bath. The cooling water kept at a temperature lower than the set temperature was circulated in the bath. The deviation corresponding to the difference between the set temperature and the observed temperature was used as the input signal. The output or control input to be given to the heater element in the water bath was calculated by the fuzzy modus. The algorithm of the output function was simplified by means of circulating the cooling water. The contribution of each membership function was changed according to the thermal progress of the observed temperature of the water. The simple control program was written in BASIC for a practical water bath, and the observed results were fairly well. Some thermal tests were also involved.

Introduction

In the heat exchange calorimetry [1], the thermal behavior between liquid and liquid or solid, such as heat of solution, can be measured simply. Sample and reference vessels are fixed differentially in a water bath. The heat evolved in the vessel is exchanged freely with the ambient water. Temperature change in the sample solution detected by the thermistor installed in each vessel is transferred to electric signals via a Wheatstone bridge, from which rate of heat evoluation and total heat effect are estimated by means of analog treatments and/or digital calculations. In the calorimetry, the arbitrary measuring temperature may be selected easily by changing the temperature of the bath water. However, the differential treatment of observed signals is necessary for estimating the rate of heat evolution, in which noisy signals lower the measurement precision. Therefore, the thermal fluctuation of the bath water should be kept within a narrow range.

For the precision water bath, the modified PID (proportional, integral and derivative) control system [2] was reported. However, most of parameters in the programs depended on the experimental conditions to be controlled, such as the volume of water and its set temperature. In addition, many constants should be changed for the empirically selected values according to the experimental conditions. Then, the problem involving parameters was improved by simulating the thermal behavior of the bath water in an exponential equation. All the optimal control inputs were calculated and selected automatically from the observed temperature of the bath water [3] which was recommended for practical use owing to the simple algorithm and the versatility for the experimental conditions.

The fuzzy inference approach was applied to select the control input for the precision water bath [4]. In this report, the control programs written in BASIC were explained for the convinience of practical uses.

Experimental

Apparatus

The heat exchange calorimeter, whose essential parts were similar to those reported previously [2, 4], was assembled. The temperature control system of the water bath is shown in Fig. 1, in which all the equipments for calorimetric measurements, such as the sample and reference vessels, the magnetic stirrers fixed to each vessel, the aluminium frame in order to fix both vessels and stirrers, and the titrant reservoirs for thermal equilibrium, were omitted for simplicity.

A glass box of $23 \times 30 \times 27\ cm^3$ in size was used as the water bath which was covered with 5 cm thick Styrofoam insulator boards, and filled with about $16\ dm^3$ of water which was agitated by a motor-driven stirrer



Fig. 1. The temperature control system.

(35 W, PS-1, Yamato Kagaku Co., Tokyo). The cooling water was kept at about 2°C lower than the given set temperature of 25°C with an electric cooler (CTE-200 and CTR-220, Komatsu Electronics Co., Tokyo), and circulated in the water bath by using a glass spiral tubing shown as C in Fig. 1 at a flow rate of 1.5 $dm^3 min^{-1}$.

Temperature of the bath water was measured by a thermistor (NLB, Shibaura Denshi Co., Tokyo) shown as Th in Fig. 1, which was composed in an arm of the Wheatstone bridge. The unbalanced voltage corresponding to the deviation T_{dev} was adjusted by a preamplifier (PM-17, Toa Dema Co., Tokyo) and acquired to a small microcomputer (M5, SORD Co., Tokyo) operated with Z-80A CPU (central processing unit), Zilog Co.. The handmade input-output interface circuits were composed of a programmable input-output (PIO) (i8255, Intel Co.), an analog-to-digital converter (ADC) (AD-80-12, Analog Devices), and a digital-to-analog converter (DAC) (DAC-80-12, Micro Network). The temperature change in the bath water of 1.5×10^{-5}°C corresponded to 1 digit. The input value of T_{dev} sent to the ADC was within the range of $\pm$2V and to the microcomputer $\pm$1999 digit. The output value of H_{out} to the power amplifier circuit was 1 V in maximum or between 0 and 1000 digit. The power amplifier circuit included power transistors and the heater element shown as H in Fig. 1. The temperature of the bath water was recorded on a $Y-t$ recorder (R-02, Rika Denki Co., Tokyo).

Algorithm and programs

The heat evolved in the vessel is exchanged freely with the surrounding bath water in the heat exchange calorimeter. Heat from motor-driven agitators and magnetic stirrers placed in the water bath also conducts into the bath water. Therefore, the temperature of the water may be elevated gradually, if it is not controlled. In the present work, the cooling water kept at a temperature lower than the set temperature was circulated in the bath water. The difference T_{dev} between the set temperature and the observed temperature was observed and acquired via the interface circuits in a microcomputer. The output value or control input H_{out} necessary for keeping the temperature of the bath water at the set temperature was calculated by means of fuzzy modus, and given to the heater element soaked in the bath water via the power amplifier circuit from the microcomputer. As only the heating may be taken into consideration, the output membership functions were simplified.

Input signal T_{dev} was in the range from -1999 to 1999, and output signal H_{out} from 0 to 1000 was given to the power amplifier circuit of the heater element. However, these signals were normalized between -1

and +1 in the calculations of fuzzy inference as T and H, respectively.

Seven fuzzy variables and the defined membership functions were used and the rule of fuzzy control was set by if-then statements in BASIC.

1) if T=NB then H=ZO
2) if T=NM then H=ZO
3) if T=NS then H=ZO
4) if T=ZO then H=ZO
5) if T=PS then H=PS
6) if T=PM then H=PM
7) if T=PB then H=PB

where NB is Negative Big, NM Negative Medium, NS Negative Small, ZO Zero, PS Positive Small, PM Positive Medium, and PB Positive Big. Input T was given by input fuzzy variables from NB to PB. Zero in T corresponded that the observed temperature was equal to the set temperature. Output H was also given by output fuzzy variables from ZO to PB. Relation of fuzzy variables and membership functions to grade G is expressed by linear functions. The T and H are indicated as the abscissa between −1 and +1, and the G of membership functions as the ordinate between 0 and 1.

All the necessary programs were composed of initialization of the control system, data acquisition, calculations of the fuzzy inference using membership functions, and output of control signals. The membership functions were selected in trial and error so as to keep smaller fluctuation range. To eliminate the offset temperature and to respond speedily to the temperature change of the bath water, the latest T and the difference ΔT or DT between the last and the last save one were taken into consideration in the three kinds of detailed fuzzy subroutines in the control program as contral 1, 2 and 3.

Procedure

The procedures of the selection of set temperature, the set of sample and reference vessels, and the preparation of water bath were followed by the manner reported previously [2, 3]. Circulation of the cooling water and execution of the temperature control program were started. The temperature of the water was monitored on the $Y-t$ recorder.

Results and Discussion

As the small microcomputer used in the system took time to calculate the output H with the BASIC interpreter of floating decimal point, sampling interval of T_{dev} was every 3 s or more. Therefore, the recovery from the fairly large thermal change might not be sufficient. However, the recoveries from thermal disturbances of ca. 160J and −120J were examined at a non-air-conditioned room by adding hot and cold water, respectively. The results were shown in Fig. 5 of the previous report [4]. The recovery time depended on cooling and heating powers of the control system, as well as the control program. The controlled temperature profile was not affected, even if the room temperature changed over a few degrees. When rotation rate of the motor-driven stirrer was decreased, the load to the motor decreased and heat flow from the motor to the bath water also decreased. The examined traces are shown in Fig. 2, in which the rotation rate was lowered in 3 steps from A to C as indicated. As expected, the temperature of the bath water lowered, and at the same time the temperature fluctuation became unstable owing to the insufficient agitation.

The fuzzy function ZO′ was added between ZO and PS to decsease fluctuations, and in the actual control ZO′ was used instead of ZO, which corresponded to relative change in the dynamic range as reported previouly [2]. All the membership functions were also modified from regular types. According to the magnitude of DT which indicates the direction of temperature change, three kinds of membership functions of H were selected as shown by B, C, and D in Fig. 3 in the previous report [4].

The influence of control on the amount of bath water was desired to examine. However, large increase or decrease in the water was impossible from the bath size. The control was not affected by the change of only



Fig. 2. The influence of rotation rate of the motor-driven stirrer on the temperature profile of fuzzy controlled water:
A; normal rate, B; lower rate, C; lowest rate.

1 dm^3 in the bath water.

The fluctuation range of the bath water controlled by the fuzzy modus was within $\pm 5 \times 10^{-4}$K or less, which was adequate range to the heat exchange calorimeter. The performance of temperature control for a water bath of laboratory use was comparable to those reported previously [2, 3]. However, if more components than T and ΔT in the present work were taken into consideration, more sophisticated control may be possibly by the fuzzy inference.

References

[1] M. Nakanishi and S. Fujieda: *Anal. Chem.*, **44** (1972) 574.
[2] S. Fujieda, M. Nakanishi and J. Kawahito: *Thermochim. Acta*, **157** (1990) 163.
[3] S. Fujieda and J. Kawahito: *Thermochim. Acta*, **190** (1991) 175.
[4] S. Fujieda and S. Tanaka: *J. Therm. Anal.*, in press.

List of the programs

Control 1

```
10 '**********************************************
20 '**  FUZZY TEMPERATURE CONTROL PROGRAMS      **
30 '**  FOR PRECISION WATER BATH                **
40 '**                                          **
50 '**  BY SHIHO TANAKA & SHUKO FUJIEDA (1991)  **
60 '**********************************************
70 '
80 '--- SORD M5 F-BASIC
90 '*******************************
100 '**  HARDWARE INITIALIZATION  **
110 '*******************************
120 '
130 '--- Input via Port B, to acquire temperature deviation
140 out &73, &9B
150 '--- Output via Port A, to send control input to the heater
160 out &63, &80
170 out &65, 225: out &61, 255
180 '
190 '**********************
200 '** INITIAL SETTING  **
210 '**********************
210 '
220 cls: print "": '--- Initial display
230 for I=1 to 37
240   locate I, 0 : print "*"
250   locate I, 22: print "*"
260 next I
270 for I=1 to 21
280   locate 1, I: print "*"
290   locate 37,I: print "*"
```

```
300 next I
310 '
320 locate 3, 3: print "Data Acquisition via SC-51 as ADC"
330 locate 10,4: print "--- +- 1999mV full scale"
340 locate  3, 7: print "Output of Control Signal via DAC"
350 locate 10, 8: print "--- 12bit ; 0 to -10V"
360 locate 5, 12: print "if END, please hit [E] key!"
370 ED$=inkey$
380 if ED$="e" or ED$="E" then goto $OWARI
390 cls
400 ' --- dimension ---
410 dim A(4), B(4), C(4), D(4), ANS1(6), ANS2(6), K(30), S(30),
    SI(30), SH(30): '--- FOR $FUZZY
420 '
430 ' --- output membership function ---
440 $DATASUU: data 10
450 $KATAMUKI: data 6, -6, 6, -6, 3, -3, 6, -6, 6, 0
460 $SEPPEN: data 2, 2, 0, 4, -1, 5, -8, 12, -10, 0, 2
470 $SITEN: data -1, 0, 0, 1, 1, 3, 4, 5, 5, 6, 6
480 $SHUTEN: data 0, 1, 1, 2, 3, 5, 5, 6, 6, 6, 6
490 restore $DATASUU: read D
500 restore $KATAMUKI
510 for I=1 to D: read K(I): next I
520 restore $SEPPEN
530 for I=1 to D: read S(I): next I
540 read S
550 for I=1 to D: S(I)=S(I)/S: next I
560 restore $SITEN
570 for I=1 to D: read SI(I): next I
580 read SI
590 for I=1 to D: SI(I)=SI(I)/I: next I
600 restore $SHUTEN
610 for I=1 to D: read SH(I): next I
620 read SH
```

```
630 for I=1 to D: SH(I)=SH(I)/SH: next I
640 '
650 '*******************
660 '** MAIN ROUTINE  **
670 '*******************
680 '
690 TM1=time: TEMPP=1999
700 gosub $SAMPLING
710 TM2=time-TM1
720 TM1=time
730 DT=TEMP-TEMPP
740 gosub $FUZZY
750 gosub $HEATER
760 TEMPP=TEMP
770 K(1)=6: K(4)=-6: S(1)=1: S(4)=2: SI(1)=-1/6: SH(4)=1/3
780 ED$=inkey$
790 if ED$="e" or ED$="E" then goto $OWARI else goto 700
800 '
810 $OWARI: end
820 '
830 '*******************
840 '** SUBROUTINES  ***
850 '*******************
860 '
870 $SAMPLING
880 '
890 INA=inp(&70): INB=inp(&71)
900 TEMP0$=mid$(hex$(INA),3,1)
910 if INA<128 then TEMP1=1 else TEMP1=0: '--- 1st digit ---
920 TEMP2=val(right$(hex$(INA),1)): '--- 2nd digit ---
930 TEMP3=val(mid$(hex$(INB),3,1)): '--- 3rd digit ---
940 TEMP4=val(right$(hex$(INB),1)): '--- 4th digit ---
950 '--- voltage via ADC (mV) ---
960 TEMP=TEMP1*1000+TEMP2*100+TEMP3*10+TEMP4
```

```
970 '--- + or - ? ---
980 if TEMPO$="0" or TEMPO$="A" or TEMPO$="B" then TEMP else TEMP=TEMP
990 return
1000 '
1010 $HEATER
1020 '
1030 VOUT%=cint(4095-(4095*(HEAT%/1000)))
1040 VOUTA1%=val("&"+mid$(hex$(VOUT%),2,1))
1050 VOUTA2%=val("&"+mid$(hex$(VOUT%),3,1))
1060 VOUTA%=VOUTA1%*16+VOUTA2%
1070 VOUTB%=val("&"+right$(hex$(VOUT%),1))
1080 out &60, VOUTA%: out &61, VOUTB%
1090 PRINT "time="; TM2; tab(9); "T="; TEMP; tab(17); "out="; HEAT%; DT;
     FUNC
1100 return
1110 '
1120 $FUZZY
1130 '
1140 if 0<=TEMP and TEMP<=30 and DT>=-5 then goto $FUNC2
1150 if 0<=TEMP and TEMP<=30 and DT<=-10 then goto $FUNC3
1160 if 0>=TEMP and TEMP>=-30 and DT<=5 then goto $FUNC3
1170 if 0>=TEMP and TEMP>=-30 and DT>=10 then goto $FUNC2
1180 if TEMP<-30 and DT<=10 then goto $FUNC3
1190 if TEMP>30 and DT>=-10 then goto $FUNC2
1200 FUNC=1
1210 goto $CALC
1220 '
1230 $FUNC2: FUNC=2
1240 K(4)=-3: S(4)=3/2: SH(4)=1/2: goto $CALC
1250 $FUNC3: FUNC=3
1260 K(1)=2: S(1)=1: SI(1)=-1/2: goto $CALC
1270 '
1280 $CALC
1290 T=TEMP/1999
```

```
1300 if -1<=T and T<-2/3 then goto $CASE1
1310 if -2/3<=T and T<-1/3 then goto $CASE2
1320 if -1/3<=T and T<0 then goto $CASE3
1330 if 0<=T and T<1/3 then goto $CASE4
1340 if 1/3<=T and T<2/3 then goto $CASE5
1350 if 2/3<=T and T<=1 then goto $CASE6
1360 '
1370 $CASE1
1380 M1=-3*T-2: M2=3*T+3
1390 A(1)=K(1)*M1: B(1)=S(1)*M1: C(1)=SI(1): D(1)=SH(1)
1400 A(2)=K(2)*M1: B(2)=S(2)*M1: C(2)=SI(2): D(2)=SH(2)
1410 A(3)=K(1)*M2: B(3)=S(1)*M2: C(3)=SI(1): D(3)=SH(1)
1420 A(4)=K(2)*M2: B(4)=S(2)*M2: C(4)=SI(2): D(4)=SH(2)
1430 ANS1=0: ANS2=1
1440 goto $OUTPUT
1450 '
1460 $CASE2
1470 M1=-3*T-1: M2=3*T+2
1480 A(1)=K(1)*M1: B(1)=S(1)*M1: C(1)=SI(1): D(1)=SH(1)
1490 A(2)=K(2)*M1: B(2)=S(2)*M1: C(2)=SI(2): D(2)=SH(2)
1500 A(3)=K(1)*M2: B(3)=S(1)*M2: C(3)=SI(1): D(3)=SH(1)
1510 A(4)=K(2)*M2: B(4)=S(2)*M2: C(4)=SI(2): D(4)=SH(2)
1520 ANS1=0: ANS2=1
1530 goto $OUTPUT
1540 '
1550 $CASE3
1560 M1=-3*T: M2=3*T+1
1570 A(1)=K(1)*M1: B(1)=S(1)*M1: C(1)=SI(1): D(1)=SH(1)
1580 A(2)=K(2)*M1: B(2)=S(2)*M1: C(2)=SI(2): D(2)=(S(2)*M1-S(3)*M2)
     /(K(3)*M2-K(2)*M1)
1590 A(3)=K(3)*M2: B(3)=S(3)*M2: C(3)=(S(2)*M1-S(3)*M2)/(K(3)*M2-K(2)
    *M1): D(3)=SH(3)
1600 A(4)=K(4)*M2: B(4)=S(4)*M2: C(4)=SI(4): D(4)=SH(4)
1610 ANS1=0
```

```
1620 for I=1 to 4
1630   gosub $SEKIBUN1
1640   ANS1=ANS1+ANS1(I)
1650 next I
1660 ANS2=0
1670 for I=1 to 4
1680   gosub $SEKIBUN2
1690   ANS2=ANS2+ANS2(I)
1700 next I
1710 goto $OUTPUT
1720 '
1730 $CASE4
1740 M1=-3*T+1: M2=3*T
1750 A(1)=K(3)*M1: B(1)=S(3)*M1: C(1)=SI(3): D(1)=SH(3)
1760 A(2)=K(4)*M1: B(2)=S(4)*M1: C(2)=SI(4): D(2)=(S(4)*M1-S(5)*M2)
     /(K(5)*M2-K(4)*M1)
1770 A(3)=K(5)*M2: B(3)=S(5)*M2: C(3)=(S(4)*M1-S(5)*M2)/(K(5)*M2
     -K(4)*M1): D(3)=SH(5)
1780 A(4)=K(6)*M2: B(4)=S(6)*M2: C(4)=SI(6): D(4)=SH(6)
1790 ANS1=0
1800 for I=1 to 4
1810   gosub $SEKIBUN1
1820   ANS1=ANS1+ANS1(I)
1830 next I
1840 ANS2=0
1850 for I=1 to 4
1860   gosub $SEKIBUN2
1870   ANS2=ANS2+ANS2(I)
1880 next I
1890 goto $OUTPUT
1900 '
1910 $CASE5
1920 M1=-3*T+2: M2=3*T-1
1930 A(1)=K(5)*M1: B(1)=S(5)*M1: C(1)=SI(5): D(1)=SH(5)
```

```
1940 A(2)=K(6)*M1: B(2)=S(6)*M1: C(2)=SI(6): D(2)=(S(6)*M1-S(7)*M2)
     /(K(7)*M2-K(6)*M1)
1950 A(3)=K(7)*M2: B(3)=S(7)*M2: C(3)=(S(6)*M1-S(7)*M2)/(K(7)*M2
     -K(6)*M1)
1960 A(4)=K(8)*M2: B(4)=S(8)*M2: C(4)=SI(8): D(4)=SH(8)
1970 ANS1=0
1980 for I=1 to 4
1990   gosub $SEKIBUN1
2000   ANS1=ANS1+ANS1(I)
2010 next I
2020 ANS2=0
2030 for I=1 to 4
2040   gosub $SEKIBUN2
2050   ANS2=ANS2+ANS2(I)
2060 next I
2070 goto $OUTPUT
2080 '
2090 $CASE6
2100 M1=-3*T+3: M2=3*T-2
2110 A(1)=K(7)*M1: B(1)=S(7)*M1: C(1)=SI(7): D(1)=SH(7)
2120 A(2)=K(8)*M1: B(2)=S(8)*M1: C(2)=SI(8): D(2)=(S(8)*M1-S(9)*M2)
     /(K(9)*M2-K(8)*M1)
2130 A(3)=K(9)*M2: B(3)=S(9)*M2: C(3)=(S(8)*M1-S(9)*M2)/(K(9)*M2
     -K(8)*M1): D(3)=SH(9)
2140 A(4)=K(10)*M2: B(4)=S(10)*M2: C(4)=SI(10): D(4)=SH(10)
2150 ANS1=0
2160 for I=1 to 4
2170   gosub $SEKIBUN1
2180   ANS1=ANS1+ANS1(I)
2190 next I
2200 ANS2=0
2210 for I=1 to 4
2220   gosub $SEKIBUN2
2230   ANS2=ANS2+ANS2(I)
```

```
2240 next I
2250 goto $OUTPUT
2260 '
2270 $OUTPUT
2280 HEAT%=(ANS1/ANS2)*1000
2290 if HEAT%<0 then HEAT%=0
2300 if HEAT%>1000 then HEAT%=1000
2310 return
2320 '
2330 $SEKIBUN1
2340 N=5
2350 S=0
2360 H=(D(I)-C(I))/N
2370 X=C(I)
2380 for J=1 to N-1
2390   X=X+H
2400   F1=A(I)*X^2+B(I)*X
2410   S=S+F1
2420 next J
2430 F2=A(I)*C(I)^2+B(I)*C(I)
2440 F3=A(I)*D(I)^2+B(I)*D(I)
2450 ANS1(I)=(F2+F3+S*2)*H/2
2460 return
2470 '
2480 $SEKIBUN2
2490 N=2
2500 S=0
2510 H=(D(I)-C(I))/N
2520 X=C(I)
2530 for J=1 to N-1
2540   X=X+H
2550   F1=A(I)*X+B(I)
2560   S=S+F1
2570 next J
```

```
2580 F2=A(I)*C(I)+B(I)
2590 F3=A(I)*D(I)+B(I)
2600 ANS2(I)=(F2+F3+S*2)*H/2
2610 return
```

Control 2

```
1120 $FUZZY
1130 '
1140 if 0<TEMP and TEMP<=30 and DT>=-5 then goto $FUNC2
1150 if 0<TEMP and TEMP<=30 and DT<=-10 then goto $FUNC3
1160 if 0>TEMP and TEMP>=-30 and DT<=5 then goto $FUNC3
1170 if 0>TEMP and TEMP>=-30 and DT>=10 then goto $FUNC2
1180 if TEMP<-30 and DT<=20 then goto $FUNC3
1190 if TEMP>30 and DT>=-20 then goto $FUNC2
1200 FUNC=1
1210 goto $CALC
```

Control 3

```
1120 $FUZZY
1130 '
1140 if 0<TEMP and TEMP<=30 and DT>=-5 then goto $FUNC2
1150 if 0<TEMP and TEMP<=30 and DT<=-10 then goto $FUNC3
1160 if 0>TEMP and TEMP>=-30 and DT<=3 then goto $FUNC3
1170 if 0>TEMP and TEMP>=-30 and DT>=10 then goto $FUNC2
1180 if TEMP<-30 and DT<=20 then goto $FUNC3
1190 if TEMP>30 and DT>=-20 then goto $FUNC2
1200 FUNC=1
1210 goto $CALC
```